Procrustes distance.sas
This routine computes the true Procrustes distance between each pair of specimens in the dataset. Procrustes distance is reported as an angle in radians, representing the geodesic distance between two specimens in Procrustes space. As written here, the routine will print pairwise distances in the SAS OUTPUT window; distances will also be stored in a dataset named “procd”. The matrix algebra for Orthogonal Procrustes Analysis comes from Rohlf & Slice (1990); the code here follows their notation as closely as possible. 
INPUT: 
One dataset containing only numeric coordinate variables and a single character variable of specimen labels. Each row should represent the coordinates of a different specimen. Additional variables can be screened out prior to beginning IML using “keep” or “drop” statements in a data step. Input data for this computation need not be superimposed. 

OUTPUT: 
Matrix of Procrustes distances between every pair of specimens 

Reference:    
Rohlf, F. J. & D. E. Slice. 1990. Extensions of the Procrustes method for the optimal superimposition of landmarks. Systematic Zoology 39:40-59.

Code annotation: 
### indicates lines that need to be changed for different datasets
proc iml;
Begins processing commands in Interactive Matrix Language.
use two; 
### Tells IML to use data from dataset “two”.
read all into X; 
Inputs ALL numeric data into a matrix “X”. 
read all var _CHAR_ into name;
Inputs ALL character variabless into a matrix “name”.
k=3; 
### k is the dimensionality of the data (either 2 or 3).
n=nrow(X); 
n is the number of specimens computed as the number of rows in matrix “X”.
p=(ncol(X)/k); 
p is the number of landmarks computed by dividing the number of columns in “X” (corresponding to the number of coordinates) by the dimensionality (k).
dmat=j(n,n);
Defines a matrix “dmat” using the j-function, with the number of rows and columns equal to the number of specimens (n). The default value for filling this matrix is 1. The purpose in generating this dummy matrix is to have a place to record the distances as they are later calculated.
y=j(1,ncol(X));
This creates a matrix “y” of one row and columns equal to the number of variables for temporary storage of specimen coordinate differences.
do f=1 to n; 
Sets up a loop with “f” cycling up to the number of specimens.
do j=1 to n;
Sets up a second loop WITHIN THE FIRST of “j”. Together these will cycle through every combination of specimen pairs.
X1=X[f,]; 
Defines X1 as the specimen given in the “f” row number.
X1=shape(X1,p,k);
Reshapes the coordinate matrix of that specimen to have p x k dimensions.
X2=X[j,]; X2=shape(X2,p,k);
These are as above with “X1”
I=I(p); 
Sets up an identity matrix “I” of rank p x p.
CaP=j(p,p,1/p);
Sets up a p x p matrix with all values equal to 1/p.
s1=sqrt(trace((I-CaP)*X1*X1`*(I-CaP)));
Computes the centroid size of the first object.
s2=sqrt(trace((I-CaP)*X2*X2`*(I-CaP)));
Computes the centroid size of the second object.
X1prime=(I-CaP)*X1/s1; 
Gives the coordinates for the first specimen translated to the origin and scaled it by its centroid size.
X2prime=(I-CaP)*X2/s2;
As above with “X1prime”
cov2=X1prime`*X2prime;
Defines cov2 as the matrix multiplication of the transpose of “X1prime” by “X2prime”.
CALL SVD (U,sig,V,cov2);
Performs a singular value decomposition on “cov2”, yielding “U”, “sig”, and “V”.
sig2=abs(sig); 
The next four lines transform “sig” into a diagonal matrix of ones with their signs corresponding to those of the original elements of “sig”. Here, “sig2” contains the absolute values for all elements of “sig”.
S=sig/sig2; 
Dividing “sig” by “sig2” yields a vector of ones keeping their original signs. 
S=S`; 
Transposes the vector.
S=diag(S);
Creates a new matrix “S” with the diagonal containing the ones from the previous “S”.
H=V*S*U`;
Defines the rotation matrix “H”.
X2star=X2prime*H;
“X2star” gives the new coordinates of “X2” translated to a common origin with “X1”, scaled to a unit centroid size, and rotated for the optimal least squares fit with “X1”. 
y=X1prime-X2star;
“y” becomes a vector of the differences between the superimposed specimens at every coordinate. 
d=sqrt(trace(y`*y));
Matrix multiplies the transpose of “y” by “y” and then takes the square root of the trace (i.e, the sum of diagonal elements which correspond to the squared differences in each coordinate) of this matrix and defines this distance as “d.”
rho=2*arsin(d/2);
Computes the angle “rho” (in radians) as twice the arcsine of d/2.
dmat[j,f]=rho;
Puts the value “rho” into the matrix “dmat.” Note that replacing “rho” with “d” here will cause the output to be a distance rather than an angle (geodesic distance).
if j=f then do; dmat[j,f]=0; end;
Automatically defines the distance between a specimen and itself as zero, eliminating potential problems with rounding error. 
end; 
Ends the “j” loop, causing “j” to cycle through again as “j”+1. 
end;
Ends the “f” loop (after all the “j”s have cycled) causing “f” to iterate again – with each iteration containing a full cycle of “j” interations. 
print dmat[rowname=name colname=name];
Prints matrix “dmat” to the output window, with row and column names gives by their specimen labels, stored earlier in “name”. 
create procd from dmat; 
Creates a SAS dataset named “procd” to record all of the pairwise Procrustes distances.
append from dmat;
Outputs “dmat” data to the dataset. 
abort;
Ends the IML session.
